# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgbase=fpylll
pkgname=python-fpylll
pkgver=0.5.9
pkgrel=2
pkgdesc='A Python interface for fplll'
arch=(loong64 x86_64)
url='https://github.com/fplll/fpylll'
license=(GPL2)
depends=(fplll python-numpy python-cysignals)
makedepends=(cython python-build python-installer python-setuptools python-wheel)
source=(https://github.com/fplll/fpylll/releases/download/$pkgver/$pkgbase-$pkgver.tar.gz)
sha256sums=('5652ce92ead07ebf402d6b85af228df106cd4c24f0c7f6cf50004722c436a248')

build() {
  cd $pkgbase-$pkgver
  python -m build --wheel --no-isolation
}

package_python-fpylll() {
  cd $pkgbase-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl
}
